


لعبة Whac-A-Mole

سنقوم في هذا المشروع ببرمجة لعبة Whac-A-mole المسلية من خلال استخدام عصا التحكم و RGB وحساب النقاط وعرضها على وحدة عرض 7-segment



الأدوات المطلوبة




Inventor Kit

القطع المستخدمة

سنستخدم في هذا المشروع القطع التالية:


		
RGB LED	حامل بطاريات 6 AA	لوحة Orion
		
عصا التحكم Joystick	7-segment	

طريقة التوصيل



القطة الالكترونية	المنفذ
RGB LED	منفذ3
segment-7	منفذ4
عصا التحكم joystick	منفذ6

الفكرة البرمجية




تحتوي وحدة RGB على 4 LED سنقوم ببرمجتها لنتم إضاءةها عشوائياً وبشكل سريع ، ونلاحق هذه الإضاءة عبر عصا التحكم، عند توافق توجيه العصا باتجاه LED المضاءة يتحول لونها من الأزرق إلى الأحمر وتسجيل نقطة تظهر على وحدة عرض 7-segment



تتحرك عصا التحكم في اتجاهات مختلفة ، نحتاج لاختبارها وتحديد القيمة التي تعطيها في كل اتجاه .

نقوم بتوصيل عصا التحكم باللوحة Orion في المنفذ 6 واستخدام الكود التالي لعرض القراءة على الشاشة



قم بتحريك عصا التحكم في اتجاه Y أعلى وأسفل وفي اتجاه X يمينا ويساراً وتسجيل القيمة التي تعطيها في كل اتجاه لنستخدم هذه القيم في برمجة اللعبة



X 128

Y -10



برمجة المشروع

سنحتاج بداية لانشاء 4 متغيرات

الوظيفة	المتغير
يشير إلى رقم LED في وحدة RGB	LED
لتسجيل النقاط عند التصويب	Score
لتخزين قيمة حركة عصا التحكم في اتجاه X	X
لتخزين قيمة حركة عصا التحكم في اتجاه y	Y

Make a Variable

- LED
- Score
- X
- Y

set LED to 0

change LED by 1

show variable LED

hide variable LED

نبدأ الكود بتعيين قيمة RGB LED وحدة segment-7 على قيمة 0
وتعيين العداد على 0

```

when clicked
  set 7-segments display Port4 number 0
  set led Port3 all red 0 green 0 blue 0
  set Score to 0

```

داخل حلقة Forever

نقوم بتعيين المتغير LED بحيث تتغير القيمة بصورة عشوائية من 0 - 4 وفق عدد LED من 0-8 لتقليل سرعة التغير في الإضاءة ثم نضع أمر إضاءة LED بحيث تضاء حسب الرقم العشوائي كل مرة

```

forever
  set LED to pick random 1 to 8
  set led Port3 LED red 0 green 0 blue 20

```

نقوم بتعيين المتغيرين X و Y على حركة عصا التحكم Joystick

```

forever
  set LED to pick random 1 to 8
  set led Port3 LED red 0 green 0 blue 20
  set X to joystick Port6 X-Axis
  set Y to joystick Port6 Y-Axis

```

نستخدم جملتين شرطيتين if..then

بحيث إذا كانت قيمة LED تساوي 1 (أي LED1 مضاءة)

وتم تصويب عصا التحكم في اتجاه LED (نضع قيمة Y وفق القيمة التي حصلنا عليها عند اختبار عصا التحكم)

```

if LED = 1 then
  if Y < -50 then

```

نضيف الأمر بحيث تتغير الإضاءة من أزرق إلى أحمر

```

set led Port3 LED red 60 green 0 blue 0

```

وتسجيل نقطة في المتغير Score

```
change Score by 1
```

وعرض عدد النقاط على وحدة 7-segment

```
set 7-segments display Port4 number Score
```

```
if LED = 1 then
  if Y < -50 then
    set led Port3 LED red 60 green 0 blue 0
    change Score by 1
    set 7-segments display Port4 number Score
```

يتم كتابة الأكواد للاتجاهات و LED الأخرى بنفس الطريقة مع وضع القيم المناسبة
قم بالإطلاع على الكود كاملا :


```
when clicked
  set 7-segments display Port4 number 0
  set led Port3 all red 0 green 0 blue 0
  set Score to 0
  forever
    set LED to pick random 1 to 8
    set led Port3 LED red 0 green 0 blue 20
    set X to joystick Port6 X-Axis
    set Y to joystick Port6 Y-Axis
    if LED = 2 then
      if X < -90 then
        set led Port3 LED red 60 green 0 blue 0
        change Score by 1
        set 7-segments display Port4 number Score
    if LED = 4 then
      if X > 100 then
        set led Port3 LED red 60 green 0 blue 0
        change Score by 1
        set 7-segments display Port4 number Score
    if LED = 3 then
      if Y > 100 then
        set led Port3 LED red 60 green 0 blue 0
        change Score by 1
        set 7-segments display Port4 number Score
    if LED = 1 then
      if Y < -50 then
        set led Port3 LED red 60 green 0 blue 0
        change Score by 1
        set 7-segments display Port4 number Score
  set led Port3 all red 0 green 0 blue 0
```

The code is a Scratch script that starts with a 'when clicked' event. It initializes a 7-segment display to 0 and sets all LEDs to off (red, green, blue = 0). A 'Score' variable is set to 0. A 'forever' loop begins, where a random LED (1-8) is selected. The LED's state is set to 'LED' (red, green, blue) with a value of 20. Joystick coordinates for X and Y are read from Port6. Four conditional blocks follow: if LED=2 and X < -90, if LED=4 and X > 100, if LED=3 and Y > 100, and if LED=1 and Y < -50. Each condition triggers setting the LED to 60, increasing the score by 1, and updating the 7-segment display. The loop ends with setting all LEDs back to off.

وللمزيد من المعلومات حول الكود يمكنك مراجعة درس جولة حول الأوامر البرمجية.